

1 **STABILIZED OPTIMIZATION VIA AN NCL ALGORITHM***

2 DING MA[†], KENNETH JUDD[‡], DOMINIQUE ORBAN[§], AND MICHAEL SAUNDERS[†]

3 **Abstract.** For optimization problems involving many nonlinear inequality constraints, we
 4 extend the BCL and LCL approaches of LANCELOT and MINOS to an algorithm that solves a
 5 sequence of nonlinearly constrained augmented Lagrangian subproblems. The effect is to regularize
 6 the nonlinear constraints by making their gradients linearly independent. The NCL algorithm is
 7 implemented in AMPL and tested on large examples of a tax policy model.

8 **1. Introduction.** We consider constrained optimization problems of the form

9

NCO	$\begin{aligned} &\text{minimize}_{x \in \mathbb{R}^n} \phi(x) \\ &\text{subject to } c(x) \geq 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \end{aligned}$
-----	--

10 where $\phi(x)$ is a smooth nonlinear function, $c(x) \in \mathbb{R}^m$ is a vector of smooth nonlin-
 11 ear functions, and $Ax \geq b$ is a placeholder for a set of linear inequality or equality
 12 constraints, with x lying between lower and upper bounds ℓ and u .

13 In some applications where $m \gg n$, there may be more than n constraints that
 14 are essentially active at a solution. The constraints do not satisfy the linear indepen-
 15 dence constraint qualification (LICQ), and general-purpose solvers are likely to have
 16 difficulty converging. Some form of regularization is required. We achieve this by
 17 adapting the augmented Lagrangian algorithm of the general-purpose optimization
 18 solver LANCELOT [2, 3, 11] to derive a sequence of regularized subproblems denoted
 19 in the next section by NC_k .

20 **2. BCL, LCL, and NCL methods.** The theory for the large-scale solver
 21 LANCELOT is best described in terms of the general optimization problem

22

NECB	$\begin{aligned} &\text{minimize}_{x \in \mathbb{R}^n} \phi(x) \\ &\text{subject to } c(x) = 0, \quad \ell \leq x \leq u \end{aligned}$
------	---

23 with *nonlinear equality constraints* and bounds. We take the primal and dual solu-
 24 tions to be (x^*, y^*, z^*) . LANCELOT treats NECB by solving a sequence of *bound-*
 25 *constrained subproblems* of the form

26

BC_k	$\begin{aligned} &\text{minimize}_x L(x, y_k, \rho_k) = \phi(x) - y_k^T c(x) + \frac{1}{2} \rho_k \ c(x)\ ^2 \\ &\text{subject to } \ell \leq x \leq u, \end{aligned}$
--------	--

27 where y_k is an estimate of the Lagrange multipliers y^* for the equality constraints.
 28 This was called a bound-constrained Lagrangian (BCL) method by Friedlander and
 29 Saunders [6] in contrast to the LCL (linearly constrained Lagrangian) methods of

*Version of December 4, 2017.
[†]Management Science and Engineering, Stanford University, Stanford, CA 94305-4026, USA
 ({dingma,saunders}@stanford.edu). Partially supported by NIH grant U01GM102098.
[‡]Hoover Institution, Stanford University, Stanford, CA 94305-6010, USA (judd@
 hoover.stanford.edu)
[§]GERAD and Department of Mathematics and Industrial Engineering, École Polytechnique,
 Montréal, QC, Canada (dominique.orban@gerad.ca). Partially supported by an NSERC Discovery
 Grant.

30 Robinson [14] and MINOS [12], whose subproblems LC_k contain bounds as in BC_k
 31 and also linearizations of the equality constraints at the current point x_k (including
 32 linear constraints).

33 In order to treat NCO with a sequence of BC_k subproblems, we convert the
 34 nonlinear inequality constraints to equalities to obtain

$$\begin{array}{ll} \text{NCO}' & \text{minimize } \phi(x) \\ & \text{subject to } c(x) - s = 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \quad s \geq 0 \end{array}$$

36 with corresponding subproblems (including linear constraints)

$$\begin{array}{ll} \text{BC}_k' & \text{minimize } L(x, y_k, \rho_k) = \phi(x) - y_k^T(c(x) - s) + \frac{1}{2}\rho_k\|c(x) - s\|^2 \\ & \text{subject to } Ax \geq b, \quad \ell \leq x \leq u, \quad s \geq 0. \end{array}$$

38 We now introduce variables $r = -(c(x) - s)$ into BC_k' to obtain the *nonlinearly*
 39 *constrained Lagrangian* (NCL) subproblem

$$\begin{array}{ll} \text{NC}_k & \text{minimize } \phi(x) + y_k^T r + \frac{1}{2}\rho_k\|r\|^2 \\ & \text{subject to } c(x) + r \geq 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \end{array}$$

41 in which r serves to make the nonlinear constraints independent. For $\rho_k > 0$ and
 42 larger than a certain finite value, the LANCELOT-type NCL algorithm should cause
 43 y_k to approach y^* and most of the solution $(x_k^*, r_k^*, y_k^*, z_k^*)$ of NC_k should approach
 44 (x^*, y^*, z^*) , with the corresponding r_k^* approaching zero.

45 Problem NC_k is analogous to Friedlander and Orban's formulation for convex
 46 quadratic programs [5, Eq. (3.2)]. See also Arreckx et al. [1].

47 Note that for general problems (NECB), the BCL and LCL subproblems contain
 48 linear constraints (bounds only, or linearized constraints and bounds). Our NCL for-
 49 mulation retains nonlinear constraints in the NC_k subproblems, but simplifies them by
 50 ensuring that they satisfy LICQ [13, Ch. 12]. On large problems, the additional vari-
 51 ables $r \in \mathfrak{R}^m$ in NC_k may be detrimental to active-set solvers like MINOS or SNOPT
 52 [7] because they increase the number of degrees of freedom (superbasic variables). For-
 53 tunately they are easily accommodated by interior methods, as our numerical results
 54 show for IPOPT [15, 8]. We trust that the same will be true for KNITRO [10].

55 **2.1. The BCL algorithm.** The LANCELOT BCL method is summarized in
 56 [Algorithm 1](#). Each subproblem BC_k is solved with a specified optimality tolerance ω_k ,
 57 generating an iterate x_k^* and the associated Lagrangian gradient $z_k^* \equiv \nabla L(x_k^*, y_k, \rho_k)$.
 58 If $\|c(x_k^*)\|$ is sufficiently small, the iteration is regarded as “successful” and an update
 59 to y_k is computed from x_k^* . Otherwise, y_k is not altered but ρ_k is increased.

60 Key properties are that the subproblems are solved inexactly, the penalty pa-
 61 rameter is increased only finitely often, and the multiplier estimates y_k need not be
 62 assumed bounded. Under certain conditions, all iterations are eventually success-
 63 ful, the ρ_k 's remain constant, the iterates converge superlinearly, and the algorithm
 64 terminates in a finite number of iterations.

65 **2.2. The NCL algorithm.** To derive a stabilized algorithm for problem NCO,
 66 we modify [Algorithm 1](#) by introducing r and replacing the subproblems BC_k by NC_k .
 67 The resulting NCL method is summarized in [Algorithm 2](#). The update to y_k becomes
 68 $y_k^* \leftarrow y_k - \rho_k(c(x_k^*) - s_k^*) = y_k + \rho_k r_k^*$, the value satisfied by an optimal y_k^* for
 69 subproblem NC_k .

Algorithm 1 BCL (Bound-Constrained Lagrangian Method for NECB)

```

1: procedure BCL( $x_0, y_0, z_0$ )
2:   Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ .
3:   Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ .
4:    $k \leftarrow 0$ , converged  $\leftarrow$  false
5:   repeat
6:      $k \leftarrow k + 1$ 
7:     Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k \rightarrow \infty} \omega_k \leq \omega_*$ .
8:     With starting point  $(x_{k-1}, z_{k-1})$ , find  $(x_k^*, z_k^*)$  that solves  $\text{BC}_k$  within tol  $\omega_k$ .
9:     if  $\|c(x_k^*)\| \leq \max(\eta_*, \eta_k)$  then
10:       $y_k^* \leftarrow y_k - \rho_k c(x_k^*)$ 
11:       $x_k \leftarrow x_k^*, y_k \leftarrow y_k^*, z_k \leftarrow z_k^*$  update solution estimates
12:      if  $(x_k, y_k, z_k)$  solves NECB, converged  $\leftarrow$  true
13:       $\rho_{k+1} \leftarrow \rho_k$  keep  $\rho_k$ 
14:       $\eta_{k+1} \leftarrow \eta_k / (1 + \rho_{k+1}^\beta)$  decrease  $\eta_k$ 
15:    else
16:       $\rho_{k+1} \leftarrow \tau \rho_k$  increase  $\rho_k$ 
17:       $\eta_{k+1} \leftarrow \eta_0 / (1 + \rho_{k+1}^\alpha)$  may increase or decrease  $\eta_k$ 
18:    end if
19:  until converged
20:   $x^* \leftarrow x_k, y^* \leftarrow y_k, z^* \leftarrow z_k$ 
21: end procedure

```

Algorithm 2 NCL (Nonlinearly Constrained Lagrangian Method for NCO)

```

1: procedure NCL( $x_0, r_0, y_0, z_0$ )
2:   Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ .
3:   Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ .
4:    $k \leftarrow 0$ , converged  $\leftarrow$  false
5:   repeat
6:      $k \leftarrow k + 1$ 
7:     Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k \rightarrow \infty} \omega_k \leq \omega_*$ .
8:     With starting point  $(x_{k-1}, r_{k-1}, y_{k-1}, z_{k-1})$ , find  $(x_k^*, r_k^*, y_k^*, z_k^*)$  that solves  $\text{NC}_k$ 
9:     within tol  $\omega_k$ .
10:    if  $\|r_k^*\| \leq \max(\eta_*, \eta_k)$  then
11:      $y_k^* \leftarrow y_k + \rho_k r_k^*$ 
12:      $x_k \leftarrow x_k^*, r_k \leftarrow r_k^*, y_k \leftarrow y_k^*, z_k \leftarrow z_k^*$  update solution estimates
13:     if  $(x_k, y_k, z_k)$  solves NCO, converged  $\leftarrow$  true
14:      $\rho_{k+1} \leftarrow \rho_k$  keep  $\rho_k$ 
15:      $\eta_{k+1} \leftarrow \eta_k / (1 + \rho_{k+1}^\beta)$  decrease  $\eta_k$ 
16:    else
17:      $\rho_{k+1} \leftarrow \tau \rho_k$  increase  $\rho_k$ 
18:      $\eta_{k+1} \leftarrow \eta_0 / (1 + \rho_{k+1}^\alpha)$  may increase or decrease  $\eta_k$ 
19:    end if
20:  until converged
21:   $x^* \leftarrow x_k, r^* \leftarrow r_k, y^* \leftarrow y_k, z^* \leftarrow z_k$ 
22: end procedure

```

k	ρ_k	η_k	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	Itns	Time
1	10^2	10^{-2}	3.1e-03	-2.1478532e+01	125	42.8
2	10^2	10^{-3}	1.3e-03	-2.1277587e+01	18	6.5
3	10^3	10^{-3}	6.6e-04	-2.1177152e+01	27	9.1
4	10^3	10^{-4}	5.5e-04	-2.1110210e+01	31	10.8
5	10^4	10^{-4}	2.9e-04	-2.1066664e+01	57	24.3
6	10^5	10^{-4}	6.5e-05	-2.1027152e+01	75	26.8
7	10^5	10^{-5}	5.2e-05	-2.1018896e+01	130	60.9
8	10^6	10^{-5}	9.3e-06	-2.1015295e+01	159	81.8
9	10^6	10^{-6}	2.0e-06	-2.1014808e+01	139	70.0
10	10^7	10^{-6}	2.1e-07	-2.1014800e+01	177	97.6

TABLE 1

NCL results on a 4D example with $na = 11$, $nb = 3$, $nc = 3$, $nd = 2$, giving $m = 39006$, $n = 396$. Itns refers to IPOPT's primal-dual interior point method, and Time is seconds on an Apple iMac with 2.93 GHz Intel Core i7.

70 **3. An application: optimal tax policy.** Some challenging test cases arise
71 from the tax policy models described in [9]. They take the form

(Tax) maximize $\sum_i \lambda_i U^i(c_i, y_i)$
subject to $U^i(c_i, y_i) - U^i(c_j, y_j) \geq 0$ for all i, j
 $\lambda^T(y - c) \geq 0$
 $c, y \geq 0,$

73 where c_i and y_i are the consumption and income of taxpayer i , and λ is a vector of
74 positive weights. The utility functions $U^i(c_i, y_i)$ are each of the form

$$75 \quad U(c, y) = \frac{(c - \alpha)^{1-1/\gamma}}{1 - 1/\gamma} - \psi \frac{(y/w)^{1/\eta+1}}{1/\eta + 1},$$

76 where w is the wage rate and α , γ , ψ and η are taxpayer heterogeneities. More
77 precisely, the utility functions are of the form

$$78 \quad U^{i,j,k,g,h}(c_{p,q,r,s,t}, y_{p,q,r,s,t}) = \frac{(c_{p,q,r,s,t} - \alpha_k)^{1-1/\gamma_h}}{1 - 1/\gamma_h} - \psi_g \frac{(y_{p,q,r,s,t}/w_i)^{1/\eta_j+1}}{1/\eta_j + 1},$$

80 where (i, j, k, g, h) and (p, q, r, s, t) run over na wage types, nb elasticities of labor
81 supply, nc basic need types, nd levels of distaste for work, and ne elasticities of demand
82 for consumption, with na , nb , nc , nd , ne determining the size of the problem, namely
83 $m = 2T$ nonlinear constraints, $n = T(T-1)$ variables, with $T := na \times nb \times nc \times nd \times ne$.

84 **Table 1** summarizes results for a 4D example (with $ne = 1$ and $\gamma_1 = 1$). The
85 first term of $U(c, y)$ becomes $\log(c - \alpha)$, the limit as $\gamma \rightarrow 1$. Problem NCO and
86 algorithm NCL were formulated in the AMPL modeling language [4]. The solvers
87 SNOPT [7] and IPOPT [15] were unable to solve NCO itself, but algorithm NCL
88 was successful with IPOPT solving the subproblems NC_k . The optimality tolerance
89 for IPOPT was $\omega_k = 10^{-6}$ throughout, and warm starts were specified for $k \geq 2$
90 (options warm_start_init_point=yes, mu_init=1e-4). These options greatly improved
91 the performance of IPOPT on each subproblem compared to cold starts, for which
92 mu_init=0.1.

93 For this 4D example, problem (NCO) has $m = 39006$ nonlinear inequality con-
94 straints and one linear constraint in $n = 395$ variables $x = (c, y)$, and nonnegativity
95 bounds. Subproblem NC_k therefore has 39007 constraints and 39402 variables when r

k	ρ_k	η_k	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	Itns	Time
1	10^2	10^{-2}	7.0e-03	-4.2038075e+02	95	41.1
2	10^2	10^{-3}	4.1e-03	-4.2002898e+02	17	7.2
3	10^3	10^{-3}	1.3e-03	-4.1986069e+02	20	8.1
4	10^4	10^{-3}	4.4e-04	-4.1972958e+02	48	25.0
5	10^4	10^{-4}	2.2e-04	-4.1968646e+02	43	20.5
6	10^5	10^{-4}	9.8e-05	-4.1967560e+02	64	32.9
7	10^5	10^{-5}	6.6e-05	-4.1967177e+02	57	26.8
8	10^6	10^{-5}	4.2e-06	-4.1967150e+02	87	46.2
9	10^6	10^{-6}	9.4e-07	-4.1967138e+02	96	53.6

TABLE 2

NCL results on a 5D example with $na = 5$, $nb = 3$, $nc = 3$, $nd = 2$, $ne = 2$, giving $m = 32220$, $n = 360$. Itns refers to IPOPT's primal-dual interior point method, and Time is seconds on an Apple iMac with 2.93 GHz Intel Core i7.

k	ρ_k	η_k	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	mu_init	Itns	Time
1	10^2	10^{-2}	5.1e-03	-1.7656816e+03	10^{-1}	825	7763.3
2	10^2	10^{-3}	2.4e-03	-1.7648480e+03	10^{-4}	66	472.8
3	10^3	10^{-3}	1.3e-03	-1.7644006e+03	10^{-4}	106	771.3
4	10^4	10^{-3}	3.8e-04	-1.7639491e+03	10^{-5}	132	1347.0
5	10^4	10^{-4}	3.2e-04	-1.7637742e+03	10^{-5}	229	2450.9
6	10^5	10^{-4}	8.6e-05	-1.7636804e+03	10^{-6}	104	1096.9
7	10^5	10^{-5}	4.9e-05	-1.7636469e+03	10^{-6}	143	1633.4
8	10^6	10^{-5}	1.5e-05	-1.7636252e+03	10^{-7}	71	786.1
9	10^7	10^{-5}	2.8e-06	-1.7636196e+03	10^{-7}	67	725.7
10	10^7	10^{-6}	5.1e-07	-1.7636187e+03	10^{-8}	18	171.0

TABLE 3

NCL results on a 5D example with $na = 21$, $nb = 3$, $nc = 3$, $nd = 2$, $ne = 2$, giving $m = 570780$, $n = 1512$. Itns refers to IPOPT's primal-dual interior point method, and Time is seconds on an Apple iMac with 2.93 GHz Intel Core i7.

is included. Fortunately r does not affect the complexity of each IPOPT iteration (but greatly improves stability). In contrast, active-set methods like MINOS and SNOPT are very inefficient on NC_k subproblems because the number of inequality constraints leads to thousands of minor iterations, and the presence of r leads to thousands of superbasic variables.

Table 2 summarizes results for a 5D example. The NC_k subproblems have $m = 32220$ nonlinear constraints and $n = 361$ variables ($\Rightarrow 32581$ variables including r). Again the options `warm_start_init_point=yes`, `mu_init=1e-4` for $k \geq 2$ led to good performance by IPOPT on each subproblem.

For much larger problems of this type, we found that it was helpful to reduce `mu_init` more often, as illustrated in Table 3. The NC_k subproblems here have $m = 570780$ nonlinear constraints and $n = 1512$ variables ($\Rightarrow 572292$ variables including r). Note that the number of NCL iterations is stable ($k \leq 10$), and IPOPT performs well on each subproblem with decreasing `mu_init`. It is helpful that only the objective function of NC_k changes as $k \leftarrow k + 1$.

111 **4. AMPL models, data, and scripts.** NCL algorithm ([Algorithm 2](#)) has been
 112 implemented in the AMPL modeling language [4] and tested on problem (Tax). The
 113 following sections list each relevant file.

114 **4.1. Tax model.** File `pTax5Dncl.mod` codes the NC_k subproblem for problem
 115 (Tax) with five parameters $w, \eta, \alpha, \psi, \gamma$, using $\mu := 1/\eta$. Note that for $U(c, y)$ in the
 116 objective and constraint functions, the first term $(c - \alpha)^{1-1/\gamma}/(1 - 1/\gamma)$ is replaced
 117 by a piecewise-continuous function that is defined for all values of c and α .

118 Primal regularization $\frac{1}{2}\pi\|(c, y)\|^2$ with $\pi = 10^{-8}$ is added to the objective function
 119 in order to promote uniqueness of the minimizer.

```

120 # pTax5Dncl.mod
121 #
122 # An NLP to solve a tax example with 5-dimensional types of tax payers.
123 #
124 # 29 Mar 2005: Original AMPL coding by K. Judd and C.-L. Su.
125 # 20 Sep 2016: Revised by D. Ma and M. A. Saunders.
126 # 08 Nov 2016: 3D version created by D. Ma and M. A. Saunders.
127 # 08 Dec 2016: 4D version created by D. Ma and M. A. Saunders.
128 # 10 Mar 2017: Switch to piece-wise utility
129 # 12 Nov 2017: pTax5Dncl.mod derived from pTax5D.mod and pTax4Dncl.mod.
130
131 # Define parameters for agents (taxpayers)
132 param na;          # number of first types
133 param nb;          # number of second types
134 param nc;          # number of third types
135 param nd;
136 param ne;
137 set A := 1..na;    # set of first types
138 set B := 1..nb;    # set of second types
139 set C := 1..nc;    # set of third types
140 set D := 1..nd;
141 set E := 1..ne;
142 set T = {A,B,C,D,E}; # set of agents
143
144 # Define wages for agents (taxpayers)
145 param wmin;        # minimum wage level
146 param wmax;        # maximum wage level
147 param w {A};       # i, wage vector
148 param mu{B};       # j, mu = 1/eta# mu vector
149 param mu1{B};      # mu1[j] = mu[j] + 1
150 param alpha{C};    # k, ak vector for utility
151 param psi{D};      # g
152 param gamma{E};    # h
153 param lambda{A,B,C,D,E};
154 param epsilon;
155 param primreg      default 1e-8;    # 1e-8;    # Small primal regularization
156
157 var c{(i,j,k,g,h) in T} >= 0.1; # consumption for tax payer (i,j,k,g,h)
158 var y{(i,j,k,g,h) in T} >= 0.1; # income      for tax payer (i,j,k,g,h)
159 var R{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
160     if i=p and j=q and k=r and g=s then h!=t} >= -1e+20, <= 1e+20;
161
162 param kmax      default 20;          # NCL limit on itn
163 param rhok      default 1e+2;        # augmented Lagrangian penalty parameter
164 param rhofac    default 10.0;        # Increase factor
165 param rhomax    default 1e+8;        # biggest rho
166
167 param etak      default 1e-2;        # opttol for augmented Lagrangian loop
168 param etafac    default 0.1;         # Reduction factor for opttol
169 param etamin    default 1e-8;        # smallest etak

```

```

170
171 param rmax      default  0;      # max r (for printing)
172 param rmin      default  0;      # min r (for printing)
173 param rnorm     default  0;      # ||r||_inf
174 param rtol      default 1e-6;    # quit if biggest r_i <= rtol
175
176 param ck{(i,j,k,g,h) in T} default 0; # current c
177 param yk{(i,j,k,g,h) in T} default 0; # current y
178 param rk{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
179   if i=p and j=q and k=r and g=s then h!=t} default 0; # r = - (c(x) - s)
180 param dk{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
181   if i=p and j=q and k=r and g=s then h!=t} default 0; # current d (duals)
182
183 minimize f:
184   sum{(i,j,k,g,h) in T}
185   (
186     (if c[i,j,k,g,h] - alpha[k] >= epsilon then
187       - lambda[i,j,k,g,h] *
188         ((c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
189         - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
190     else
191       - lambda[i,j,k,g,h] *
192       (- 0.5/gamma[h] * epsilon^(1-1/gamma[h]-1) * (c[i,j,k,g,h] - alpha[k])^2
193       + (1+1/gamma[h])* epsilon^(1-1/gamma[h]) ) * (c[i,j,k,g,h] - alpha[k])
194       + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h]) * epsilon^(1-1/gamma[h])
195       - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
196     )
197   + 0.5 * primreg * (c[i,j,k,g,h]^2 + y[i,j,k,g,h]^2)
198   )
199 + sum{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
200   if i=p and j=q and k=r and g=s then h!=t}
201   (dk[i,j,k,g,h,p,q,r,s,t] * R[i,j,k,g,h,p,q,r,s,t]
202   + 0.5 * rhok * R[i,j,k,g,h,p,q,r,s,t]^2);
203
204 subject to
205
206 Incentive {(i,j,k,g,h) in T, (p,q,r,s,t) in T:
207   if i=p and j=q and k=r and g=s then h!=t}:
208   (if c[i,j,k,g,h] - alpha[k] >= epsilon then
209     (c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
210     - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
211   else
212     - 0.5/gamma[h] * epsilon^(1-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
213     + (1+1/gamma[h])*epsilon^(1-1/gamma[h])*(c[i,j,k,g,h] - alpha[k])
214     + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
215     - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
216   )
217 - (if c[p,q,r,s,t] - alpha[k] >= epsilon then
218   (c[p,q,r,s,t] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
219   - psi[g]*(y[p,q,r,s,t]/w[i])^mu1[j] / mu1[j]
220   else
221     - 0.5/gamma[h] * epsilon^(1-1/gamma[h]-1)*(c[p,q,r,s,t] - alpha[k])^2
222     + (1+1/gamma[h])*epsilon^(1-1/gamma[h])*(c[p,q,r,s,t] - alpha[k])
223     + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
224     - psi[g]*(y[p,q,r,s,t]/w[i])^mu1[j] / mu1[j]
225   )
226   + R[i,j,k,g,h,p,q,r,s,t] >= 0;
227
228 Technology:
229   sum{(i,j,k,g,h) in T} lambda[i,j,k,g,h]*(y[i,j,k,g,h] - c[i,j,k,g,h]) >= 0;

```

230 **4.2. Tax model data.** File pTax5Dnc1.dat provides data for a specific problem
 231 case.

```

232 # pTax5Dnc1.dat
233
234 data;
235
236 let na := 21;
237 let nb := 3;
238 let nc := 3;
239 let nd := 2;
240 let ne := 2;
241
242 # Set up wage dimension intervals
243 let wmin := 2;
244 let wmax := 4;
245 let {i in A} w[i] := wmin + ((wmax-wmin)/(na-1))*(i-1);
246
247 param mu :=
248     1  0.5
249     2  1
250     3  2 ;
251
252 # Define mu1
253 let {j in B} mu1[j] := mu[j] + 1;
254
255 data;
256
257 param alpha :=
258     1  0
259     2  1
260     3  1.5;
261
262 param psi :=
263     1  1
264     2  1.5;
265
266 param gamma :=
267     1  2
268     2  3;
269
270 # Set up 5 dimensional distribution
271 let {(i,j,k,g,h) in T} lambda[i,j,k,g,h] := 1;
272
273 # Choose a reasonable epsilon
274 let epsilon := 0.1;

```


275 **4.3. Initial values.** File `pTax5Dinitial.run` solves a simplified model to com-
 276 pute starting values for Algorithm NCL. This model solves easily with MINOS or
 277 SNOPT on all cases tried. Solution values are output to file `p5Dinitial.dat`.

```

278 # pTax5Dinitial.run
279
280 # Define parameters for agents (taxpayers)
281 param na := 21;      # number of types in wage
282 param nb := 3;      # number of types in eta
283 param nc := 3;      # number of types in alpha
284 param nd := 2;      # number of types in gamma
285 param ne := 2;      # number of types in psi
286 set A := 1..na;     # set of first types
287 set B := 1..nb;     # set of second types
288 set C := 1..nc;     # set of third types
289 set D := 1..nd;
290 set E := 1..ne;
291 set T = {A,B,C,D,E}; # set of agents
292
293 # Define wages for agents (taxpayers)
294 param wmin := 2;    # minimum wage level
295 param wmax := 4;    # maximum wage level
296 param w {i in A} := wmin + ((wmax-wmin)/(na-1))*(i-1); # wage vector
297
298 # Choose a reasonable epsilon
299 param epsilon := 0.1;
300
301 # mu vector
302 param mu {B};      # mu = 1/eta
303 param mu1{B};      # mu1[j] = mu[j] + 1
304 param alpha {C};
305 param gamma {E};
306 param psi {D};
307
308 var c {(i,j,k,g,h) in T} >= 0.1;
309 var y {(i,j,k,g,h) in T} >= 0.1;
310
311 maximize f: sum{(i,j,k,g,h) in T}
312   if c[i,j,k,g,h] - alpha[k] >= epsilon then
313     (c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
314     - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
315   else
316     - 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
317     + (1+1/gamma[h])*epsilon^(-1/gamma[h]) *(c[i,j,k,g,h] - alpha[k])
318     + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
319     - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j];
320
321 subject to
322
323   Budget {(i,j,k,g,h) in T}:
324     y[i,j,k,g,h] - c[i,j,k,g,h] = 0;
325
326
327 let {(i,j,k,g,h) in T} y[i,j,k,g,h] := i+1;
328 let {(i,j,k,g,h) in T} c[i,j,k,g,h] := i+1;
329
330 data;
331
332 param mu :=
333   1  0.5
334   2  1
335   3  2 ;

```

```
336
337 # Define mu1
338 let {j in B} mu1[j] := mu[j] + 1;
339
340 data;
341
342 param alpha :=
343     1  0
344     2  1
345     3  1.5;
346
347 param psi :=
348     1  1
349     2  1.5;
350
351 param gamma :=
352     1  2
353     2  3;
354
355 option solver minos;
356 option solver snopt;
357 option show_stats 1;
358
359 option minos_options ' \
360     summary_file=6      \
361     print_file=9       \
362     scale=no           \
363     print_level=0      \
364     *minor_iterations=200 \
365     major_iterations=2000 \
366     iterations=50000   \
367     optimality_tol=1e-7 \
368     *penalty=100.0     \
369     completion=full    \
370     *major_damp=0.1    \
371     superbasics_limit=3000 \
372     solution=yes      \
373     *verify_level=3   \
374 ';
375
376 option snopt_options ' \
377     summary_file=6      \
378     print_file=9       \
379     scale=no           \
380     print_level=0      \
381     major_iterations=2000 \
382     iterations=50000   \
383     optimality_tol=1e-7 \
384     *penalty=100.0     \
385     superbasics_limit=3000 \
386     solution=yes      \
387     *verify_level=3   \
388 ';
389
390
391 display na,nb,nc,nd,ne;
392 solve;
393 display na,nb,nc,nd,ne;
394 display y,c >p5Dinitial.dat;
395 close p5Dinitial.dat;
```

396 **4.4. Implementation of algorithm NCL.** File `pTax5Dnclipopt.run` uses

```

pTax5Dinitial.run
pTax5Dncl.mod
397 pTax5Dncl.dat
pTax5Dinitial.dat

```

398 to implement Algorithm NCL. Subproblems NC_k are solved in a loop until $\|r_k^*\|_\infty \leq$
399 $\text{rtol} = 1\text{e-}6$, or η_k has been reduced to parameter $\text{etamin} = 1\text{e-}8$, or ρ_k has been
400 increased to parameter $\text{rhomax} = 1\text{e+}8$. The loop variable k is called K to avoid a
401 clash with subscript k in the model file.

402 Optimality tolerance $\omega_k = 10^{-6}$ is used throughout to ensure that the solution of
403 the final subproblem NC_k will be close to a solution of the original problem if $\|r_k^*\|_\infty$
404 is small enough for the final k ($\|r_k^*\|_\infty \leq \text{rtol} = 1\text{e-}6$).

405 IPOPT is used to solve each subproblem NC_k , with runtime options set to im-
406 plement increasingly warm starts.

```

407 # pTax5Dnclipopt.run
408
409 reset;
410 model pTax5Dinitial.run;
411 reset;
412 model pTax5Dncl.mod;
413 data pTax5Dncl.dat;
414 data; var include p5Dinitial.dat;
415
416 model;
417 option solver ipopt;
418 option show_stats 1;
419
420 option ipopt_options ' \
421   dual_inf_tol=1e-6   \
422   max_iter=5000      \
423 ' ;
424
425 # NCL method.
426 # kmax, rhok, rhofac, rhomax, etak, etafac, etamin, rtol
427 # are defined in the .mod file.
428
429 printf "NCLipopt log for pTax5D\n" > 5DNCLipopt.log;
430 display na, nb, nc, nd, ne, primreg > 5DNCLipopt.log;
431 printf "  k      rhok      etak      rnorm      Obj\n" > 5DNCLipopt.log;
432
433 for {K in 1..kmax}
434 { display na, nb, nc, nd, ne, primreg, K, kmax, rhok, etak;
435   if K == 2 then
436   {option ipopt_options $ipopt_options
437     ' warm_start_init_point=yes   \
438     mu_init=1e-4                   \
439   ' };
440   if K == 4 then {option ipopt_options $ipopt_options ' mu_init=1e-5'};
441   if K == 6 then {option ipopt_options $ipopt_options ' mu_init=1e-6'};
442   if K == 8 then {option ipopt_options $ipopt_options ' mu_init=1e-7'};
443   if K ==10 then {option ipopt_options $ipopt_options ' mu_init=1e-8'};
444
445   solve;
446
447   let rmax := max({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
448     if i=p and j=q and k=r and g=s then h!=t} R[i,j,k,g,h,p,q,r,s,t]);
449   let rmin := min({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
450     if i=p and j=q and k=r and g=s then h!=t} R[i,j,k,g,h,p,q,r,s,t]);

```

```

451  display na, nb, nc, nd, ne, primreg, K, rhok, etak, kmax;
452  display K, kmax, rmax, rmin;
453  let rnorm := max(abs(rmax), abs(rmin)); # ||r||_inf
454
455  printf "%4i %9.1e %9.1e %9.1e %15.7e\n", K, rhok, etak, rnorm, f >> 5DNCLipopt.log;
456  close 5DNCLipopt.log;
457
458  if rnorm <= rtol then
459  { printf "Stopping: rnorm is small\n"; display K, rnorm; break; }
460
461  if rnorm <= etak then
462  { let {(i,j,k,g,h) in T, (p,q,r,s,t) in T:
463      if i=p and j=q and k=r and g=s then h!=t}
464      dk[i,j,k,g,h,p,q,r,s,t] :=
465      dk[i,j,k,g,h,p,q,r,s,t] + rhok*R[i,j,k,g,h,p,q,r,s,t];
466      let {(i,j,k,g,h) in T} ck[i,j,k,g,h] := c[i,j,k,g,h];
467      let {(i,j,k,g,h) in T} yk[i,j,k,g,h] := y[i,j,k,g,h];
468      display K, etak;
469      if etak == etamin then { printf "Stopping: etak = etamin\n"; break; }
470      let etak := max(etak*etafac, etamin);
471      display etak;
472  }
473  else # keep previous solution
474  { let {(i,j,k,g,h) in T} c[i,j,k,g,h] := ck[i,j,k,g,h];
475      let {(i,j,k,g,h) in T} y[i,j,k,g,h] := yk[i,j,k,g,h];
476      display K, rhok;
477      if rhok == rhomax then { printf "Stopping: rhok = rhomax\n"; break; }
478      let rhok := min(rhok*rhofac, rhomax);
479      display rhok;
480  }
481 }
482
483 display c,y;
484 display na, nb, nc, nd, ne, primreg, rhok, etak, rnorm;
485 printf "Created 5DNCLipopt.log\n";

```

486 **5. Conclusions.** This work has been illuminating in several ways as we sought
487 to improve our ability to solve examples of problem (Tax).

- 488 • Small examples of the tax model solve efficiently with MINOS and SNOPT,
489 but eventually fail to converge as the problem size increases.
- 490 • IPOPT also solves small examples efficiently, but eventually starts requesting
491 additional memory for the MUMPS sparse linear solver. The solver may
492 freeze, or the iterations may diverge.
- 493 • It is often said that interior methods cannot be warm-started. Nevertheless,
494 IPOPT has several runtime options that have proved to be extremely helpful
495 for implementing Algorithm NCL. For the results obtained here, it has been
496 sufficient to say that warm starts are wanted for $k > 1$, and that the IPOPT
497 barrier parameter should be initialized at smaller and smaller values as the
498 objective of subproblem NC_k changes with k .

499 **Acknowledgments.** We are extremely grateful to the developers of AMPL and
500 IPOPT for making the development and evaluation of Algorithm NCL possible. We
501 are especially grateful to Mehi Al-Baali and other organizers of the NAO-IV conference
502 *Numerical Analysis and Optimization* at Sultan Qaboos University, Muscat, Oman,
503 which brought the authors and AMPL developers together in January 2017.

504

REFERENCES

- 505 [1] S. Arreckx, D. Orban, and N. van Omme. NLP.py: An object-oriented environment for large-
506 scale optimization. Technical Report GERAD G-2016-42, GERAD, Montréal, QC, Canada,
507 2016.
- 508 [2] A. R. Conn, N. I. M. Gould, and Ph. L. Toint. A globally convergent augmented Lagrangian
509 algorithm for optimization with general constraints and simple bounds. *SIAM J. Numer.*
510 *Anal.*, 28:545–572, 1991.
- 511 [3] A. R. Conn, N. I. M. Gould, and Ph. L. Toint. *LANCELOT: A Fortran Package for Large-*
512 *scale Nonlinear Optimization (Release A)*. Lecture Notes in Computation Mathematics 17.
513 Springer Verlag, Berlin, Heidelberg, New York, London, Paris and Tokyo, 1992.
- 514 [4] R. Fourer, D. M. Gay, and B. W. Kernighan. *AMPL: A Modeling Language for Mathematical*
515 *Programming*. Brooks/Cole, Pacific Grove, second edition, 2002.
- 516 [5] M. P. Friedlander and D. Orban. A primal–dual regularized interior-point method for convex
517 quadratic programs. *Math. Prog. Comp.*, 4(1):71–107, 2012.
- 518 [6] M. P. Friedlander and M. A. Saunders. A globally convergent linearly constrained Lagrangian
519 method for nonlinear optimization. *SIAM J. Optim.*, 15(3):863–897, 2005.
- 520 [7] P. E. Gill, W. Murray, and M. A. Saunders. SNOPT: An SQP algorithm for large-scale con-
521 strained optimization. *SIAM Review*, 47(1):99–131, 2005. SIGEST article.
- 522 [8] IPOPT open source optimization solver. <https://projects.coin-or.org/Ipopt>.
- 523 [9] K. L. Judd, D. Ma, M. A. Saunders, and C.-L. Su. Optimal income taxation with multidimen-
524 sional taxpayer types. Working paper, Hoover Institution, Stanford University, 2017.
- 525 [10] KNITRO optimization software. https://www.artelys.com/tools/knitro_doc/2_userGuide.html.
- 526 [11] LANCELOT optimization software. <http://www.numerical.rl.ac.uk/lancelot/blurb.html>.
- 527 [12] B. A. Murtagh and M. A. Saunders. A projected Lagrangian algorithm and its implementation
528 for sparse nonlinear constraints. *Math. Program. Study*, 16:84–117, 1982.
- 529 [13] J. Nocedal and S. J. Wright. *Numerical Optimization*. Springer Series in Operations Research.
530 Springer Verlag, New York, second edition, 2006.
- 531 [14] S. M. Robinson. A quadratically-convergent algorithm for general nonlinear programming prob-
532 lems. *Math. Program.*, 3:145–156, 1972.
- 533 [15] A. Wächter and L. T. Biegler. On the implementation of a primal-dual interior point filter line
534 search algorithm for large-scale nonlinear programming. *Math. Program.*, 106(1), 2006.